TEMPLATE = lib
CONFIG += qt plugin
QT += qml quick
TARGET = Graphics

DESTDIR = ../build/graphics

HEADERS += \ 
	ShapeItem.h \
	PointItem.h \
	LineItem.h \
	CircleItem.h \
	RectItem.h \
	BezierCurveItem.h \
	GraphicsPlugin.h

SOURCES += \ 
	ShapeItem.cpp \
	PointItem.cpp \
	LineItem.cpp \
	CircleItem.cpp \
	RectItem.cpp \
	BezierCurveItem.cpp \
        GraphicsPlugin.cpp


OTHER_FILES += qmldir

!equals(_PRO_FILE_PWD_, $$DESTDIR) {
    copy_qmldir.target = $$DESTDIR/qmldir
    copy_qmldir.depends = $$_PRO_FILE_PWD_/qmldir
    copy_qmldir.commands = $(COPY_FILE) \"$$replace(copy_qmldir.depends, /, $$QMAKE_DIR_SEP)\" \"$$replace(copy_qmldir.target, /, $$QMAKE_DIR_SEP)\"
    QMAKE_EXTRA_TARGETS += copy_qmldir
    PRE_TARGETDEPS += $$copy_qmldir.target
}

qmldir.files = qmldir
unix {
    installPath = $$[QT_INSTALL_QML]/$$replace(uri, \\., /)
    qmldir.path = $$installPath
    target.path = $$installPath
    INSTALLS += target qmldir
}
